# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=dependent-sum-template
pkgname=haskell-dependent-sum-template
pkgver=0.1.1.1
pkgrel=124
pkgdesc="Template Haskell code to generate instances of classes in dependent-sum package"
url="https://github.com/obsidiansystems/dependent-sum"
license=("custom:PublicDomain")
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-dependent-sum' 'haskell-th-abstraction' 'haskell-th-extras')
makedepends=('ghc' 'haskell-constraints-extras')
source=("https://hackage.haskell.org/packages/archive/$_hkgname/$pkgver/$_hkgname-$pkgver.tar.gz")
sha256sums=('d46fc18d5c2f5d385f901777e982e8a1ae9094ba7e68f7e4997a82036375ae0f')

build() {
  cd $_hkgname-$pkgver

  # Tests are broken: https://github.com/obsidiansystems/dependent-sum/pull/57#issuecomment-865277051
  runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
    --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --disable-tests \
    --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm \
    --ghc-option=-optl-Wl\,-z\,relro\,-z\,now \
    --ghc-option='-pie'

  runhaskell Setup build $MAKEFLAGS
  runhaskell Setup register --gen-script
  runhaskell Setup unregister --gen-script
  sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
  sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
  cd $_hkgname-$pkgver
  #runhaskell Setup test --show-details=direct
}

package() {
  cd $_hkgname-$pkgver

  install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
  install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
  runhaskell Setup copy --destdir="$pkgdir"
}
